



CHAPTER

6

Examples That Use Remote SQL Pass-Through (RSPT)

Example 1. RSPT Services: Querying a Table in DB2 51

Purpose 51

Program 51

Example 2. RSPT Services: Subsetting Remote SAS Data 52

Purpose 52

RLS Program 52

RSPT Program 53

Example 1. RSPT Services: Querying a Table in DB2

Purpose

This example shows how to query a DB2 table that is located on a remote host by using SQL statements issued from a local host SAS session.

Program

The following sequence of statements would be used in an OS/390 SAS session to establish a connection to DB2 and query the table SYSIBM.SYSTABLES:

```
connect to db2 (ssid=db2p);

select * from connection to db2
  (select name, creator, colcount
   from sysibm.systables
   where creator='THOMPSON' or
         creator='JONES');
```

The same connection and query could be performed from an OS/2 SAS session by using RSPT by means of a SAS server in a remote session under OS/390:

```
connect to remote
  (server=rmt dbms=db2 dbmsarg=(ssid=db2p));
select * from connection to remote
  (select name, creator, colcount
   from sysibm.systables
   where creator='THOMPSON' or
         creator='JONES');
```

Use the AS alias clause in the CONNECT TO statement to give the same name to the connection to the remote DBMS as it would have if you connected directly to it. This enables you to use queries without changing the FROM CONNECTION TO clause:

```
connect to remote as db2
  (server=rmt dbms=db2 dbmsarg=(ssid=db2p));

select * from connection to db2
  (select name, creator, colcount
   from sysibm.systables
   where creator='THOMPSON' or
         creator='JONES');
```

Example 2. RSPT Services: Subsetting Remote SAS Data

Purpose

The PROC SQL view SALES97 presents sales data for fiscal year 1997 and is defined on a UNIX workstation as follows:

```
create view servlib.sales97 as
  select sum(amount) as amount
  from sales
  where year=1997;
```

Processing this view (by using RLS from your local SAS session under Windows) is comparatively fast because the view is interpreted in the server SAS session. The summary function SUM() is applied when the view is interpreted and only the summary row is returned to your local SAS session.

If you want to obtain only your own sales data and break down the sales by customer, you could use RLS or RSPT. The following sections show you how either of these services could be used and explain why RSPT is a better choice.

RLS Program

You can create a new view in your local SAS library to access the underlying data by using RLS from your local SAS session under Windows, as follows:

```
libname mylib 'C:\sales';

libname servlib '/dept/sales/revenue'
  server=servername;

create view mylib.sales97 as
  select customer, sum(amount) as amount
  from servlib.sales
  where year=1997 and
        salesrep='L. PETERSON'
  group by customer
  order by customer;
```

However, processing this view is expensive because the summary is not performed until the data reaches the local SAS session. This means more data is sent across the

network. In the following RSPT example, the summary is done before data is transferred. This reduces the amount of data that crosses the network.

RSPT Program

The following statements create a new PROC SQL view in a local SAS library that uses RSPT to access the remote SAS data:

```
libname mylib 'C:\sales';

libname servlib '/dept/sales/revenue'
  server=servername;

proc sql;
connect to remote
  (server=servername);

create view mylib.sales97 as
  select * from connection to remote
    (select customer, sum(amount) as amount
     from servlib.sales
     where year=1997 and
           salesrep='L. PETERSON'
     group by customer
     order by customer);
```

Note: The libref SERVLIB must be defined for the remote SAS library either in your SAS session or in the server SAS session. In this example, a LIBNAME statement is executed in the local SAS session to access the library through the server that is running in the remote session. Alternatively, you could remote submit a LIBNAME statement to define the library. Δ

You may want to create a view in the remote server, which can be used by many people. By modifying the previous example to include all sales reps, the view satisfies the needs of users who are interested in the sales made by more than one sales rep. The following example creates a view in the server session that summarizes the data by customer for all sales reps:

```
libname servlib '/dept/sales/revenue'
  server=servername;

proc sql;
connect to remote
  (server=servername);

execute
  (create view servlib.cust97 as
   select customer,
   sum(amount) as amount from sales
   where year=1997
   group by customer) by remote;
```


The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/CONNECT User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. pp. 537.

SAS/CONNECT User's Guide, Version 8

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-477-2

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, September 1999

SAS[®] and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. [®] indicates USA registration.

IBM[®], AIX[®], DB2[®], OS/2[®], OS/390[®], RS/6000[®], System/370[™], and System/390[®] are registered trademarks or trademarks of International Business Machines Corporation. ORACLE[®] is a registered trademark or trademark of Oracle Corporation. [®] indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.